
This is the page to save portpins for far more relevant tasks than just sensing some 
keys. The trick used is to connect keys, switches and keypads to an AD converter in
put and to leave the task to identify the ones pressed to a short piece of software.

If you need more explanations on the resistor matrix's function and calculation, con
sult the numerous subpages. 

Attaching and reading of several keys and switches via an ADC input pin can be done 
in different ways. Three options are described here, depending from your hardware: 

Hardware Page

Whatever hardware you have, this fits it all Link

If you have up to eight single keys of which only one 
of it can be pressed at a time

Link

If you have up to four switches, each of which can be 
on and off simultaneously (such as a DIP switch or a 
mouse piano)

Link

 

If you have a key matrix with either 12 keys in a 3-
by-4 or 16 keys in a 4-by-4 matrix, where only one 
of the keys can be pressed at a time, and if you 
want to learn how to interface that with an AVR

Link

... and you prefer using a command line tool to 
iterate resistor values

Link

... and you prefer using a graphical application tool 
to iterate resistor values

Link

file:///C:/Users/gerd/Documents/9_websites/gsc-da/html/avr-asm/avr_en/apps/key_matrix/keypad/rmatrix/rmatrix.html
file:///C:/Users/gerd/Documents/9_websites/gsc-da/html/avr-asm/avr_en/apps/key_matrix/key_stack/key_stack.html
file:///C:/Users/gerd/Documents/9_websites/gsc-da/html/avr-asm/avr_en/apps/key_matrix/keypad/resmatrix/resmatrix.html
file:///C:/Users/gerd/Documents/9_websites/gsc-da/html/avr-asm/avr_en/apps/key_matrix/keypad/keyboard.html


Software for all purposes
As calculating resistors for keys, switches and keypads is pretty much the same for all 
cases,  this (for Win64) software in Lazarus Pascal has been written to calculate all 
variants.  

It can be used as follows. 

1. From the drop-down Type select either keys, switches or keypads,

2. depending from that selection, the Size drop-down offers specific sizes,

3. choose resistor and ADC properties, if desired,

After having selected all properties you can press Iterate1 for single iterations or It
erate100 for 100 iterations.

The read-only edit  field  Difference 
shows the average difference of the 
resulting  voltages  from  the  target 
voltages.  The  edit  field  Overlaps 
provides information on the number 
of  overlapping  areas  between  the 

different keys and states of the input keys and switches. Steps counts the total num
ber of iteration steps,  Success the number of successful iterations. If you'd like to 
restart iteration choose Restart.

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/multi_adc.zip


From the group Show

you can open different windows that display the current iteration status and results:

1. Resistors opens a window displaying the current resistor values and their lower 
and upper values, depending from the current tolerance setting,

If you klick on a value in that window the 
content is saved as a text file, see the file 
as an example. 

2. Schematic opens a graphic window displaying the electrical schematic

If you click on that picture the image is saved 
as a Portable Network Graphic (.PNG) file or a 
Bitmap Graphics (.BMP) file. 

3. Results displays  a  window holding  all  results  for  all  different  states  (keys, 
switch combinations), with the target value, the nominal value, the lower and 
upper bound (all in mV/V) as well as the nominal, lower and upper ADC values, 
depending from the selected ADC resolution,

If you click on one of the values displayed the results are saved either as Tab 
Separated Values (.TSV) or as Comma Separated Values (.CSV), ready to be 
imported to a spreadsheet (such as Calc in the Open-Office package, just select 
Open with in the context menue of the TSV file and Calc, or, in Excel, open an 
empty spreadsheet and drop the file onto that). Note that in case of the TSV file 
the decimal separator character is taken from the operating system properties 
so that comma-separated float values are correctly recognized. 

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/Keypad_4-by-4_E12_1%25_8-bit.txt


4. Asm source provides an assembler table which can be imported into assembler 
source code,

If you click on that window its con
tent is saved as an Assembler In
clude  file  (.INC),  ready  to  be  in
cluded with  .include "filename.inc" 
into your source code. 

5. All resp. None selects resp. unselects all the windows. 

The filename is in all cases "[type]_[size]_[resistor row]_[resistor tolerance%]_[ADC 
resolution]". The location/path can be selected in the dialog.

1 Keys and resistors
If your program needs to read in two or more keys and has to detect which one of the 
several keys is pressed you can attach each key to the switch and to one input pin, 
set the internal pull-up resistor and wait for a low to occur at any of those pins. But 
you can solve this task in a very much simpler way: by switching together all your 
keys and by generating a specific voltage for each of the pressed keys. Only one sin
gle pin, an AD converter input pin, is necessary for that. 

1.1 Single voltage dividers

In this scheme, each key pressed switches a voltage di
vider on. If no key is pressed, the voltage is zero. The 
voltages produced by pressed keys are

Un = Uop * R0 / (R0 + Rn)

The resistors  R1 to  RN have 

to be different to produce dif
ferent  voltages.  And  they 
should be selected to provide 
as-far-as-possible  different 
voltages depending from the 
number N of total keys. 

1.2 Stacked keys and resistors

This is a possible variation that differs only in the mode of 
calculation.  

Here is how you can stack your keys, associated with sev
eral  resistors.  In  case  that  no  key  is  pressed,  the  ADC 
senses zero or one on this input. If one of the attached 
keys is pressed the resistors form a voltage divider: 

• The resistor R0 is the lower part of the divider. 
• Depending from the key pressed, the resistors above the pressed key add up. If 

key 1 is pressed R1, R2, ..., RN add up and the resistor series is attached to the 

positive voltage. 

The resulting voltage is



U = Uop * R0 / (R0 + R1 + R2 + ... + RN)

Depending from the operation mode of the ADC is 8 or 10 bits wide (8 bits = ADLAR 
set and read high byte of the AD result), the resulting count is either between 0..255 
or between 0..1023.

You can either use a wild bunch of resistors and measure the AD result to identify the 
key-specific readings. Simpler is it to calculate those in advance and independent. 

2 Algorithm for selecting resistors
In order to have as-high-as-possible differences between the voltages of  the keys 
those should be equally distributed over the whole available range of voltages. When 
2 keys are attached, the voltage of the first key should be the operating voltage divid
ed by two. If three keys are attached the first two keys shall be at the 0.33-fold resp. 
0.66-fold of the operating voltage. If ten keys are attached, the first voltage shall be 
the 0.111-fold of the operating voltage.

Unfortunately, real resistors are not exact but have a tolerance. This can be +/-5% for 
standard carbon resistors or +/-1% for metal film resistors. Therefore the calculated 
voltages all have a range. This ranges from 

• the lowest voltage, when R0 is on its lowest tolerance (-5%) and all other resis

tors are on their upper tolerance (+5%), to 
• the highest voltage, when R0 is on its upper range and all other resistors are on 

their lower tolerance (-5%). 

Further, resistors are only available in standardized rows. For each decade either 12 
different values are available (E12 row) or 24 (E24 row). A little more exotic are 48 
values per decade (E48), but those are not generally available.

Because of all that it is not recommendable to solve this by up to ten equations. It is  
more appropriate to 

1. vary randomly selected resistors and to determine 
2. whether a smaller value or a larger resistor value reduces the sum of differ

ences between all target and effective values. 

By minimizing the differences the number of overlaps between the upper bound volt
age of one key and the lower bound voltage of the next key also reduces. If the num
ber of such overlaps does not reduce to zero, 2% or 1% resistor values or resistors of 
the E24 row can be used.

 



3 Software for optimizing resistor values
The software is, for Windows operat
ing systems or for Linux's Wine, the 
the executable code is here (906 kB 
zipped, 2.82 MB unzipped).

At  startup,  two  keys  are  selected. 
Use  the  drop-down  field  to  have 
more.  Check  the  other  drop-down-
fields for more options. Note that all 
selections reset the resistor values to 
1k,  so  that  iteration  has  to  be 
restarted.

Iteration can be done steps by step 
or with 100 steps at once. If the dif
ferences do not reduce any more, no 
successful  steps  have  been  made 
any more. If you are not lucky with 
the result, restart and repeat the it
eration process.

The  results  of  the  iteration  can  be 
viewed in the resistor window and in 
the values window. The resistor win
dow shows the currently selected re
sistors. If those do not change after 
a single steps has been performed, 
the iteration step did not result in a 
smaller difference sum. Because the 
resistor is selected randomly, further iterations might lead to a different result. The ta
ble on the bottom shows the targeted voltages, the nominal voltages and the lower 
and upper bounds.

If you click on the button "Source 
code" an assembler table is creat
ed  that  holds  the  minima  and 
(maxima+1) of the selected keys. 
You can save this as an include file 
or  copy  it  to  the  clipboard  and 
paste this into your source code.  

The entries in that table are as fol
lows: 

1. If the ADC result is below the first entry, no key has been pressed. 
2. If the ADC result is larger or equal the first entry and smaller than the second 

entry, the respective key is pressed. 
3. If the result is equal or above the second entry and smaller than the next first 

entry, the voltage is in between two valid voltage ranges and no valid key is 
pressed, an error results. 

4. If the first table entry is zero, the last key is valid and the conversion is com
plete. 

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/key_stack/Nkey.zip


If the button "Schematic" is pressed a window opens 
displaying a graphics file with the schematic. The file 
can be saved in PNG or in BMP format by clicking on 
it.

The resistor values in the stacked format show that 
anything else than equal values are appropriate to 
achieve an appropriate voltage distribution. 

4 Assembler source code for identifying pressed keys
The following code 

1. sets the AD multiplexer to channel 0 and selects ADLAR (8 bit ADC), 
2. switches the ADC on and starts a conversion, 
3. waits for the ADC to complete, 
4. reads in the ADC result and 
5. converts the result to find the key pressed. 

The assembler source code (here in asm format) is as follows: 

;
; ***********************************
; * Decode keys                     *
; * (C)2018 by avr-asm-tutorial.net *
; ***********************************
;
.Nolist
.include "tn13def.inc" ; ATtiny13
.list
;
; -----------------------
;    R E G I S T E R S
; -----------------------
;
.def rAdcH = R14 ; High byte ADC result
.def rTab = R15 ; Key table value
.def rmp = R16 ; Multipurpose register
; Used: Z = R31:R30 for LPM
;
Start:
  ; Get an ADC value from channel 0
  ldi rmp,1<<ADLAR ; Left-adjust, channel = 0
  out ADMUX,rmp
  ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0)
  out ADCSRA,rmp ; Start ADC, start conversion

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/key_stack/decode_key.asm


WaitAdc:
  sbic ADCSRA,ADSC ; Wait for ADC complete
  rjmp WaitAdc ; Wait on
  in rAdcH,ADCH ; Read MSB
  ; Convert ADC value to key pressed
  ldi ZH,High(2*Keytab) ; Point Z to key table
  ldi ZL,Low(2*Keytab)
  clr rTab ; No key attached?
IdentifyKey:
  lpm rmp,Z+ ; Read first byte from table
  tst rmp ; Last byte in table?
  breq KeyNIdentified
  cp rAdcH,rmp ; Compare ADC value with table byte
  brcs KeyUnderflow ; ADC value below
  inc rTab ; Next key
  lpm rmp,Z+ ; Read upper bound of key value
  cp rAdcH,rmp ; Compare voltage
  brcs KeyIdentified ; Key found
  rjmp IdentifyKey ; Go on table check
;
; Last key identified
KeyNIdentified:
  inc rTab ; Key = N
  rjmp KeyIdentified
;
; No key pressed or illegal voltage
KeyUnderflow:
  tst rTab ; Key = 0?
  breq KeyIdentified ;  No key pressed
  clr rmp
  dec rmp ; Illegal voltage, rmp at 0xFF
;
KeyIdentified:
  ; The key number is in rTab
  rjmp KeyIdentified ; Loop forever
;
; Include the value table
;
.include "Key10_single.inc" ; Include the key table
;

The included table, as generated with the software, is as follows (here is the source 
code file): 

;
; Table for recognizing pressed keys
;   10 keys, ADC resolution = 256
;   R0 = 1k
;
KeyTab:
.db 25,31 ; Key=1, R1=8k2
.db 48,57 ; Key=2, R2=3k9
.db 74,86 ; Key=3, R3=2k2
.db 96,109 ; Key=4, R4=1k5
.db 121,135 ; Key=5, R5=1k
.db 146,159 ; Key=6, R6=680
.db 178,189 ; Key=7, R7=390
.db 197,206 ; Key=8, R8=270
.db 225,231 ; Key=9, R9=120
.db 0,0 ; No more keys
;

When  simulating  program  execution  with 
avr_sim yields the following results.

The conversion has been started with an analog 
voltage of 2.5 V. Conversion takes some time be
cause the selected ADC clock is slow. 

http://www.avr-asm-tutorial.net/avr_sim/index_en.html
http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/key_stack/Key10_single.inc
http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/key_stack/Key10_single.inc


The result  of  the conversion, 0x7F or  127 decimal, 
has been read to the register rAdcH in R14. 

After executing the conversion routine the assembler 
code found that the fifth key has been pressed (in 
rTab or R15). 

1 DIP switches
Rather often your AVR has to read in DIP switch
es to react on user input. If you want to read in a 
DIP  switch  with  four  single  switches  you  need 
four I/O pins, with pull-up resistors on, and the 
DIP switches pull the pin low if they are closed. 

With the scheme to 
the  left,  the  num
ber of necessary pins can be reduced from 4 down to 1 
if an AD converter input channel measures the voltage 
produced by the resistors.

A warning at the early beginning: do not try this with 
more than four switches, it does not work! Rather use 
two AD channels for 6 or 8 bit switches.

It is clear that the resistors have to be different: each 
switch has to produce roughly the double  voltage in
crease than its previous one. Unfortunately the math of 
parallel resistors is a little bit strange:

Rhigh = 1 / (1 / R1 + 1 / R2 + ... + 1 / RN)

And the resulting voltage of
UADC = Uop * R0 / (R0 + Rhigh)

is anything else than straight-forward and linear math.

So selecting the optimal resistors is a task for a large table calculation or, better, a dy
namic iteration task. 

2 Optimizing resistors
Optimization has to consider that 

1. resistor values are organized in rows. The E12 row provides 12 different values 
er decade (1.0, 1.2, 1.5, ..., 8.2) while E24 has 24 values (1.0, 1.1, 1.2, 1.3, 
1.5, ..., 9.1) per decade. Theoretically also E48 and E96 are defined but are not 



held in stock by electronic shops. 
2. resistor values are not exact due to production tolerances. Available tolerances 

are +/-5%, +/-2% and +/-1%. If you measure the resistor value manually you 
can achieve less tolerance but this is unsuitable for more than only a few de
vices. 

3. the ADC can measure in two modes: 8 bit or 10 bit. Smaller inaccuracies re
quire extra chips and are more subject to noise, so better forget this. 

This  software,  written  in 
Lazarus Pascal and executable 
under  Windows  and  Linux 
(Wine)  operating  systems, 
performs the lengthy iteration 
processes.

You can select 

1. the  number  of  DIP 
switches (2, 3 or 4), 

2. the  resistor  row to  be 
selected  from  (E12, 
E24 or E48 - if you find 
a trader offering that), 

3. the  resistor  tolerance 
(5%, 2%, 1% or 0.5% 
and  0.1% if  you  want 
to hand-select them), 

4. the ADC's resolution (8 
or 10 bit). 

The  button  "Restart"  resets 
the  resistor  values  to  their 
default. Note that each itera
tion  sequence  might  end  up 
with  different  results  as  the 
resistor to be checked next is 
randomly selected.

By pushing the button "Iter
ate1"  a  resistor  between  1 
and  N  is  randomly  selected. 
For  the  next  resistor  in  the 
row,  for  the  previous  resistor  in  the  row and  for  its  current  value  the  averaged 
squared sum of differences between the voltage required and the resulting voltage is 
calculated. If the previous or the next resistor value produce less differences those are 
selected. A window appears that displays the resistors and the resulting differences, 
so you can follow the iteration process.

The button "Iterate100" repeats this 100 times without displaying interim results. 

The resistors are displayed in the upper right corner. If you click onto a resistor line, 
you can change its value manually by clicking the "Change Rn" button. Resistor values 
are selected from the currently selected E row only. 

The result window on the bottom displays 

1. the target value of the voltage (in mV/V), 
2. the nominal value of the voltage (resistors = nominal resistance), 
3. the  lowest  and  the  highest  voltage  value  taking  resistor  tolerances  into 

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/mousepiano/mousepiano.zip


account), 
4. the nominal and the lowest and highest ADC readings for the selected resolu

tion. 

Note that the lower and upper voltages are calculated with all resistors at their highest 
tolerance (low: R0 the smallest, all other R the highest) so that you can risk to switch 
tolerance one level higher than the resistors are, because all four are probably not at 
their upper value. 

The button "Source code" writes an AVR assembler table like this (include source code 
is here): 

;
; Table for recognizing mouse piano state
;   4 switches, ADC resolution = 1024
;   Resistors E24, tolerance = 0.5%
;     R0 = 4k7
;     R1 = 68k
;     R2 = 27k
;     R3 = 10k
;     R4 = 2k4
;
PianoTab:
.dw 66,68 ; N=1
.dw 150,154 ; N=2
.dw 199,203 ; N=3
.dw 325,330 ; N=4
.dw 356,362 ; N=5
.dw 398,404 ; N=6
.dw 423,429 ; N=7
.dw 675,680 ; N=8
.dw 683,688 ; N=9
.dw 694,700 ; N=10
.dw 701,707 ; N=11
.dw 722,728 ; N=12
.dw 728,734 ; N=13
.dw 737,742 ; N=14
.dw 742,747 ; N=15
.dw 0,0 ; No more stages
;
; (+)
;  |   __    / S4
;  o--|R4|--O O--
;  |   --        |
;  |   __    / S3|
;  o--|R3|--O O--o
;      --        |
;  |   __    / S2|
;  o--|R2|--O O--o
;      --        |
;  |   __    / S1|
;   --|R1|--O O--o--> ADC input
;      --        |
;                -
;               | |R0
;               |_|
;                |
;               ---
;

The button "Schematic" draws a graphic of the cur
rent results. By clicking on it it can be saved as a 
PNG or BMP file. 

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/mousepiano/piano4.inc


AVR-Single-Chip-
Processors AT90S, 

ATtiny, ATmega
of ATMEL in practical examples.

Connecting a keypad to an AVR
This page shows how to connect a common 12-key-keypad to an AVR and read it by 
assembler software. The chapters are: 

1. How a keypad works 
2. AVR: I/O-connection of the matrix 
3. AVR: Connecting to an ADC with a resistor network 

1. How a keypad works
Keypads are switches, that are connected to rows and columns. 
If key "1" is pressed, the column 1 is electrically connected to 
row 1, if key "2" is pressed column 2 with row 1, and so on ...
To find out, if  any one of  the 12 
keys is pressed, it  would only be 
necessary to tie the three column 
lines  to  ground,  connect  all  four 
row  lines  and,  via  a  resistor  of 
10 kΩ, pull them to the operating 
voltage. If no key is pressed, the 
row lines are at plus. Any pressed 

key pulls down the row lines to zero volts.
To detect, which one of the 12 keys is pressed, pull down 
the three column lines one by one to ground (the other 
two column lines to plus) and read the resulting four row-
lines.  If  one  of  the  four  row-lines  is  low,  stop  further 
reading and identify the key code from the column and 
row info. Like that:

Column Row Key

Col1 Col2 Col3 Row1 Row2 Row3 Row4 Character Binary code

0 0 0 1 1 1 1 (none) 1111

0 1 1 0 1 1 1 1 0001

1 0 1 0 1 1 1 2 0010

1 1 0 0 1 1 1 3 0011



Column Row Key

Col1 Col2 Col3 Row1 Row2 Row3 Row4 Character Binary code

0 1 1 1 0 1 1 4 0100

1 0 1 1 0 1 1 5 0101

1 1 0 1 0 1 1 6 0110

0 1 1 1 1 0 1 7 0111

1 0 1 1 1 0 1 8 1000

1 1 0 1 1 0 1 9 1001

0 1 1 1 1 1 0 * 1010

1 0 1 1 1 1 0 0 0000

1 1 0 1 1 1 0 # 1011

Reading such a keypad using digital logic components, you need at least: 

• an oscillator, a shift register and a start/stop gate for generating the column 
signals, 

• detection if one of the four row signals is zero, 
• a recoder for conversion of the seven signals to the keycode. 

Or a complete IC doing all that (probably you won't get such an IC at your local elec
tronic parts dealer). Or you better use a micro.

2. AVR: I/O-connection of the matrix
A keypad matrix can be connected directly to an I/O port of 
an  AVR,  with  no  additional  components.
The example shows a connection to the lower seven I/O pins 
of Port B. Other ports can be used similarly. The port pins 
PB4..PB6 are defined as outputs, they provide the column 
signals. The port pins PB0..PB3 are used to read in the row 
results.  Pull-up  resistors  on  these  inputs  are  enabled  by 
software, external resistors are not necessary.

The following software example first demonstrates initializa
tion of the ports. This software part has to be executed only 
once at the AVR's program start.

Init-routine

;
; Init keypad-I/O
;
.DEF rmp = R16 ; define a multipurpose register
; define ports
.EQU pKeyOut = PORTB ; Output and Pull-Up-Port
.EQU pKeyInp = PINB  ; read keypad input
.EQU pKeyDdr = DDRB  ; data direction register of the port
; Init-routine
InitKey:



        ldi rmp,0b01110000 ; data direction register column lines output
        out pKeyDdr,rmp    ; set direction register
        ldi rmp,0b00001111 ; Pull-Up-Resistors to lower four port pins
        out pKeyOut,rmp    ; to output port

Check for any key pressed

The following routine detects if any one of the 12 keys is pressed. This routine is 
called in intervals, e.g. in a delay loop or by use of a timer. 

;
; Check any key pressed
;
AnyKey:
        ldi rmp,0b00001111 ; PB4..PB6=Null, pull-Up-resistors to input lines
        out pKeyOut,rmp    ; of port pins PB0..PB3
        in rmp,pKeyInp     ; read key results
        ori rmp,0b11110000 ; mask all upper bits with a one
        cpi rmp,0b11111111 ; all bits = One?
        breq NoKey         ; yes, no key is pressed

Identify the key pressed

Now the keypad is read out. One after the other the port bits PB6, PB5 and PB4 are 
set to low, and PB0..PB3 are checked for zeros. The register pair Z (ZH:ZL) points to a 
table with the key codes. When leaving the routine, this pair points to the key code of 
the pressed key. By using the LPM instruction, the key code is read from the table in 
the flash memory to the register R0. 

;
; Identify the key pressed
;
ReadKey:
        ldi ZH,HIGH(2*KeyTable) ; Z is pointer to key code table
        ldi ZL,LOW(2*KeyTable)
        ; read column 1
        ldi rmp,0b00111111 ; PB6 = 0
        out pKeyOut,rmp
        in rmp,pKeyInp ; read input line
        ori rmp,0b11110000 ; mask upper bits
        cpi rmp,0b11111111 ; a key in this column pressed?
        brne KeyRowFound ; key found
        adiw ZL,4 ; column not found, point Z one row down
        ; read column 2
        ldi rmp,0b01011111 ; PB5 = 0
        out pKeyOut,rmp
        in rmp,pKeyInp ; read again input line
        ori rmp,0b11110000 ; mask upper bits
        cpi rmp,0b11111111 ; a key in this column?
        brne KeyRowFound ; column found
        adiw ZL,4 ; column not found, another four keys down
        ; read column 3
        ldi rmp,0b01101111 ; PB4 = 0
        out pKeyOut,rmp
        in rmp,pKeyInp ; read last line
        ori rmp,0b11110000 ; mask upper bits
        cpi rmp,0b11111111 ; a key in this column?
        breq NoKey ; unexpected: no key in this column pressed
KeyRowFound: ; column identified, now identify row
        lsr rmp ; shift a logic 0 in left, bit 0 to carry
        brcc KeyFound ; a zero rolled out, key is found
        adiw ZL,1 ; point to next key code of that column
        rjmp KeyRowFound ; repeat shift
KeyFound: ; pressed key is found 
        lpm ; read key code to R0
        rjmp KeyProc ; continue key processing
NoKey:
        rjmp NoKeyPressed ; no key pressed
;
; Table for code conversion
;



KeyTable:
.DB 0x0A,0x07,0x04,0x01 ; First column, keys *, 7, 4 and 1
.DB 0x00,0x08,0x05,0x02 ; second column, keys 0, 8, 5 and 2
.DB 0x0B,0x09,0x06,0x03 ; third column, keys #, 9, 6 and 3

Debouncing

The routines KeyProc and NoKeyPressed have to de-bounce the pressed key. For ex
ample by counting a counter up whenever the same key is identified. Repeat this for 
e.g.  50  milliseconds.  The  NoKeyPressed  routine  clears  the  counter  and  the  key 
pressed. Because the timing depends on other necessary timing requirements of the 
AVR program, it is not shown here. 

Hints, Disadvantages

The routines shown above do not leave more time between the setting of the column 
address and the reading of  the row information.  At high clock frequencies and/or 
longer connections between keypad and processor it is necessary to leave more time 
between write and read (e.g. by inserting NOP instructions). The internal pull-ups re
sistors have values around 50 kΩ. Long lines or a noisy environment might interfere 
and produce glitches. If you like it less sensitive, add four external pull-ups.

A disadvantage of the circuit is that it requires seven port lines exclusively. The modi
fication with an AD converter and a resistor network (see chapter 3) is more economic 
and saves port lines. 

3. Connection to an ADC with a resistor matrix
Most of the Tiny and Mega-AVR devices nowadays have an AD converter on board. 
Without additional external hardware these are capable of measuring analog voltages 
and resolve these with 10 bit resolution. Those who want to save I/O ports only have 
to get the keypad to produce an analog voltage. That's the task for a resistor matrix.

Hint: An improved version of this text with more examples, a command line software 
tool etc.  is available on page  27ff! A graphical software tool for studying different 
scheme versions and sizes can be found on page 23ff. 

Resistor matrix

This is such a resistor matrix. The columns are con
nected to ground, in between the column connec
tions are three stacked resistors. The rows are con
nected via four such stacked resistors to the operat
ing voltage (e.g.  5 V).  The AD converter  input is 
blocked by a condensator of 1 nF because the ADC 
doesn't like high frequencies, that could be caught 
by the keys, the resistors and the more or less long 
lines in between all this. If the key "5" is pressed, a 
voltage divider gets active:

• 1 k + 820 Ω = 1,82k to ground,

• 3,3 k  +  680 Ω  +  180 Ω  =  4,16k  to  plus.

At an operating voltage of 5 Volt a divided voltage of
5 * 1,82 / (1,82 + 4,16) = 1,522 Volt

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/rmatrix/rmatrix.html

is seen on the AD converter input. If we consider 5% tolerance of the resistors, the 
resulting voltage is somewhere between 1,468 and 1,627 Volts. The 10-bit AD con
verter converts this (at 5 V reference voltage) to a value between 300 and 333. If we 
ignore the lower two bits of the result (e.g. divide the AD result by four or left-adjust
ing the result - if the ADC provides that function) this yields an 8-bit-value between 
74 and 78.

Each key pressed produces a typical voltage range, to be converted to the key code. 

Key encoding table

The matrix uses resistors of the E12 row, which are available at every electronic store. 
That is why the resistor values (e.g. 820 or 3300) are a little bit funny numbers.

The different resistors are necessary to achieve a smooth curve of analog voltages 
from those 12 different keys. If you think that it is more practical to just use one val
ue for all, try to calculate the resulting voltages. You will soon learn that all these re
sistors have to have different values. The reason for that is in part that they add up in 
the two rows, another reason is the voltage divider formula, which is not linear. If you 
are able to solve that in an optimal way with a few formulas, try that. I gave up very 
soon.

The following is a table for the above displayed matrix that calculates the resulting re
sistors for each key pressed. Of course, if you assume that two keys can be pressed at 
once, you have to move to a PC keyboard instead. But not many people press two 
keys at once.

The second row of the table displays the resistor values towards ground, the third to
wards the operating voltage. The voltage that results from that voltage dividers is cal
culated with the formula 

U = Uref * RGND / (RGND + RVCC) 

Because available resistors have a tolerance of +/-5% the minimum and maximum 
has to be calculated. To calculate the minimum, all resistors to ground are assumed to 
have -5%, all towards VCC are assumed to have +5%. That is not very realistic, but 
we do not want to hand-select resistors and test them in summer and winter.

Resistor divider Voltages ADC values

Key
Resistor
to GND
Ohm

Resistor
to VCC
Ohm

U Nominal
@5V

U Min
@5V

U Max
@5V

N Nom
@8Bit

N Min
@8Bit

N Max
@8Bit

1 1000 19160 0.248 0.225 0.273 13 11 14

2 1820 19160 0.434 0.396 0.475 22 20 25

3 2820 19160 0.641 0.588 0.700 33 30 36

4 1000 4160 0.969 0.893 1.050 50 45 54

5 1820 4160 1.522 1.418 1.630 78 72 84

6 2820 4160 2.020 1.901 2.142 103 97 110

7 1000 860 2.688 2.563 2.812 138 131 144

8 1820 860 3.396 3.285 3.503 174 168 180

9 2820 860 3.832 3.740 3.919 196 191 201



Resistor divider Voltages ADC values

Key
Resistor
to GND
Ohm

Resistor
to VCC
Ohm

U Nominal
@5V

U Min
@5V

U Max
@5V

N Nom
@8Bit

N Min
@8Bit

N Max
@8Bit

* 1000 180 4.237 4.170 4.300 217 213 221

0 1820 180 4.550 4.507 4.589 233 230 235

# 2820 180 4.700 4.671 4.727 241 239 243

The voltages that are associated with the keys, and their variation through tolerance 
is displayed in the following graph.

As can be seen, the tolerance influences only the middle keys.

In the table, the values N are added that would result from an AD conversion with 8-
Bit. This resolution is fine as all the values have enough distance from the previous 
and the next value. 

Voltages and key recognition

As can be seen from the table, there is no overlapping of the different detection val
ues for the keys, taking 5% tolerance of the resistors into account.

Those who like to play around with other resistor combinations, can download the cal
culation sheet (as Open-Office-Format, as Excel-XP-Format). 

Hints for the AD converter hardware

ATtiny devices in the most cases provide only the opportunity to use an internally gen
erated voltage or the supply voltage of the AVR as reference for the AD converter. For 
the keypad conversion only the supply voltage is suitable as reference. This option has 
to be selected when initiating the AD converter hardware at program start. 

Many ATmega types can connect the reference voltage to an external pin, AREF. This 

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resistor-encoder.xls

pin can either be input or output. It is an output if either the supply voltage or the in
ternal reference are selected as AD converter reference. In this case the AREF pin 
should have a condensator to ground to further reduce noise on the reference voltage. 
The AREF pin is an input, if an external reference source is selected as option. In this 
case an external source provides the reference voltage. If an external source provides 
the reference voltage, the keypad matrix should also be supplied by this source. Note 
in that case that the keypad consumes up to 10 mA, to improve noise sensitivity.

ATmega devices allow to supply the AD converter from an extra pin (AVCC) to further 
reduce noise. If only the keypad uses AD conversion the low necessary resolution of 8 
bits does not require a separate supply for the AVCC pin, it can be tied to the normal 
supply voltage. If other measuring tasks have to be performed on other channels, the 
AVCC pin should be connected to the supply voltage via a coil of 22 µH and should be 
blocked by a condenser of 100 nF to ground. 

Initiation and reading the AD converter result

For reading the keypad matrix voltage one AD converter channel is required. The AD 
converter is initiated once during program start. The two example codes show a start 
sequence for single conversion, here for an ATmega8, and one for an interrupt driven 
automatic start of the ADC, here for an ATtiny13. 

ATmega8: manual start of the ADC

The first example shows a routine for an ATmega8, without interrupts, with a manual 
start and stop of the the AD converter. The keypad signal is connected to AD channel 
ADC0. 

.DEF rKey = R15 ; Register for AD value

.DEF rmp = R16 ; Multi purpose register
        ; set MUX to channel 0, left adjust the result, AREF taken from AVCC
        ldi rmp,(1<<REFS0)|(1<<ADLAR) ; ADMUX channel 0, AREF from AVCC
        out ADMUX,rmp
        ; switch AD conversion on, start conversion, divider rate = 128
        ldi rmp,(1<<ADEN)|(1<<ADSC)|(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0)
        out ADCSRA,rmp
        ; wait until the AD conversion is complete
WaitAdc1:
        ; check ADSC bit, conversion complete if bit is zero
        sbic ADCSRA,ADSC ; conversion ready?
        rjmp WaitAdc1 ; not yet
        ; read MSB of the AD conversion result
        in rKey,ADCH
        ; switch AD converter off
        ldi rmp,(1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0) ; switch ADC off
        out ADCSRA,rmp

Please note that this single conversion requires 25 * 128 clock cycles, at 1 Mcs/s clock 
3.2 milli-seconds. Only do this that way, running around in circles, if you don't need to 
care for other things in between this time delay (except if these other things are done 
inside interrupts). 

ATtiny13: Autostart AD conversion, interrupt-driven

Yes, an ATtiny13 with its 8 pins can read our keypad matrix (we cannot connect the 
keypad matrix itself due to the limited number of I/O pins).

A typical routine for this task would be the following sequence, that converts the volt
age on ADC3 (pin 2 of the ATtiny13) continuously (after conversion complete, the next 
conversion is started automatically). 



;
; Start AD converter
;
        ; PB3=ADC3 is used for the conversion
        ldi rmp,0b00001000 ; disconnect the PB3 digital driver, saves supply current
        out DIDR0,rmp
        ; Reference = supply voltage, Left-adjust the result
        ;   ADMUX to ADC3
        ldi rmp,0b00100011 ; reference voltage = supply voltage, chose ADC3
        out ADMUX,rmp
        ; select autostart option
        ldi rmp,0b00000000 ; free-running conversion (autostart)
        out ADCSRB,rmp
        ; start ADC, allow interrupt, select clock divider
        ldi rmp,0b11101111 ; start ADC, autostart,
        out ADCSRA,rmp ;  Int Enable, clock divider to 128
; initiation complete

Running in interrupt mode requires defining the respective int jump vector, e.g. 

;
; Reset and int vectors, ATtiny13
;
.CSEG ; assemble into code segment
.ORG $0000 ; at the beginning of the code segment
        rjmp main ; Reset vector
        reti ; Int0 interrupt vector
        reti ; PCINT0 vector
        reti ; TC0 overflow vector
        reti ; Eeprom ready vector
        reti ; Analog comparator int vector
        reti ; TC0 CompA vector
        reti ; TC0 CompB vector
        reti ; WDT vector
        rjmp intadc ; ADC conversion complete vector
;

Of course, the stack must be initiated to use interrupts, and the interrupt status flag 
must be set (SEI).

The service routine intadc reads the AD conversion result. Because Left-Adjust has 
been selected, it is sufficient to read the MSB of the result: 

;
; Interrupt Service Routine AD conversion
;
.DEF rKey = R15 ; result register for the conversion result
intadc:
        in rKey,ADCH ; read AD converter MSB
        reti ; return from interrupt
;

The register rKey continously provides the current value of the resistor matrix. 

Converting the AD result to the key code

The conversion result is, as such, not very useful. The voltages and the conversion re
sult do not follow easy mathematical laws (the resistor values 4.7 - 5.6 - 6.8 - 8.2 
must  have  been  designed  by  a  drunken  math  professor,  the  formula 
V = R1 / (R1 + R2) is not very easy to handle), so that we better use a table to re
solve our key codes. The table cannot be a primitive look-up table, because we have 
256 different possible results of the conversion, and we like more slim tables.

Like a monkey, we climb the matrix tree by going step by step through the following 
table: 

KeyTable:
.DB 7, 255, 18, 1, 28, 2, 42, 3, 64, 4, 91, 5
.DB 121, 6, 156, 7, 185, 8, 207, 9, 225, 10, 237, 0, 255, 11



The first byte is the compare value for our conversion result, the second byte is the 
key code, if this compare value is greater than our result. If the result is between 0 
and <7: no key is pressed (key code is 255), if it is between 7 and <18 the key code 
is 1, etc. 

Or, if you prefer ASCII for the key codes: 

KeyTable:
.DB 7, 0 , 18, '1', 28, '2', 42, '3', 64, '4', 91, '5'
.DB 121, '6', 156, '7', 185, '8', 207, '9', 225, '*', 237, '0', 255, '#'

The code for key translation goes like this: 

;
; Converting a AD result to a key code
;
GetKeyCode:
        ; if the AD result can change in between, the result must be copied first
        mov R1,rKey ; copy AD result to R1
        ldi ZH,HIGH(2*KeyTable) ; Z points to conversion table
        ldi ZL,LOW(2*KeyTable)
GetKeyCode1:
        lpm ; read one table value
        cp R1,R0 ; compare AD result with table value
        brcs GetKeyCode2 ; less than table value, key identified
        inc R0 ; test, if table end is reached
        breq GetKeyCode2 ; reached end of table
        adiw ZL,2 ; point to next table entry
        rjmp GetKeyCode1 ; go on comparing next entry
GetKeyCode2:
        adiw ZL,1 ; point to MSB = key code
        lpm ; read key code to R0
;

Of course we have to check, if no key is pressed (R0 = 0xFF resp. if ASCII: R0 = 0) 
and we have to check for glitches (if the same key code comes 20 or more times, I 
take it for serious ...). 

Experiences

The hard- and software work very reliable. In the first version the resistor values of 
the matrix were ten times higher. This version was more vulnerable to HF noise, e.g. 
when transmitting with a 2 W VHF transmitter nearby.



AVR-Single-Chip-
Processors AT90S, 

ATtiny, ATmega
of ATMEL in practical examples.

Software for a 1 to 16 key wide keypad on 
an AVR
This page shows the software for connecting keypads of between 1 and 16 keys to an 
AVR. 

1. the software, 
2. how it works, 
3. what output it produces, 
4. which experiences have been made, 
5. feedback. 

1 The 
software
The software 
is  written  in 
Delphi-Pas
cal.  The  zip
ped  execut
able  for 
Win32  is 
here.

The following 
sections  of 
the  window 
are visible on 
the left side: 

• selection of the keypad column and row size, between 1 by 1 and 4 by 4, 
• selection of the resistor row to be used (E12 and E24 are practicable, E48 and 

E96 more theoretical)  and their  tolerance (5% and 1% are practical,  2% is 
more theoretical), 

• selection of the ADC resolution (8 bits for smaller sizes up to 12 keys, 10 bits 
for 16 keys), 

• the current scheme with the keys and the resistors, 

On the right side of the window 

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/rmatrix/RMat.zip


• the  ap
proxi
mation 
steps 
can  be 
select
ed, and 

• the cur
rent  re
sults 
are  dis
played. 

The  Win32 
software  is 
also  exe
cutable  under 
Wine in Linux, 
and looks very 
much  alike 
there. The only difference is the font of the result window.

2 How it works

2.1 Selecting the keypad size

With the column and row drop-down selectors the size of the keypad can be selected. 
Please note that selection of the size restarts all resistor values to their default value 
set. This can be used to restart the approximation from default values if an approxi
mation run ends with an unfavorable result (count of overlaps larger than zero). 

2.2 Selecting scheme variations

Two variations can be made to the scheme: 

• The "Rmode" selector allows either stacked or single resistors. This affects R5 to 
R8 in the 4-by-4 matrix (see the different schemes displayed). 

• The "Ymode" selector allows either inclusive or exclusive placement of the out
put pin. Inclusive means that the lower resistors (R2 to R4 in the 4-by-4 ma
trix) add up to the lower resistor value, while exclusive means that R1 alone is 
the lower resistor value and that R2 to R4 (in the 4-by-4 version) add up to the 
upper resistor value (see the different schemes). 

2.3 Resistor type and ADC resolution

The resistor row and ADC resolution is selected automatically from the keypad size 
but can be changed manually to understand the resulting effects. Reducing the E row 
down can result in changes of resistors because resistor values are rounded to their 
nearest equivalent in the selected E row. Changing the ADC resolution affects the 
number of overlapping voltage areas, the result table and the generated assembler ta
ble (see below).



2.4 Approximation steps

The button "Step" performs a single approximation step. Approximation means that 

• a certain resistor above R1 is randomly selected, 
• the current distances between the reference and nominal values for all  keys 

(see the Ref and Nom readings for all keys in the Result-Voltages field) is calcu
lated and compared to the distances resulting from decreasing the resistor val
ue to the next lower row value and to the distances resulting from increasing its 
value to the next higher row value, 

• the resistor with the smallest distance is then selected (either the lower, the 
current or the higher value), 

• if the resistor value has been changed in that step, it is displayed in red in the 
scheme. 

The button "Run low" performs 100 approximation steps at once only on the lower re
sistors (R2 to R4 in the 4-by-4 case), the button "Run all" includes all resistors above 
R1. The number of steps performed at once can be changed in the editor field, 100 
should be sufficient in most cases. All resistors that changed their value during the 
last run are displayed in red. The total number of runs is displayed in the respective 
label in the lower left field, together with the distances and the overlap count. The 
overlap count is increased in all cases where the Max value of a key is equal or larger 
than the Min value of the next key. It should be zero in a successful approximation 
run.

As the resistors are randomly chosen, each run can result in different resistor value 
combinations. If the overlap count is not zero and no resistor value changed, just try a 
new approximation by selecting the same column or row size. 

2.5 Changing resistor values manually

By clicking on a resistor in the scheme a window opens that allows to change its val
ue. k, K and M are allowed. Clicking on the button "Set" changes the value according
ly.

3 What it produces
If the last step or run yielded the same distance between reference and nominal val
ues and zero overlaps, the three buttons "Save as ...", "Save" and "Asm table" are en
abled. By pressing the button 

• "Save as ..." the scheme is saved either as bitmap or as a gif graphic file, 
• "Save" the resistor list is saved as a plain text file, 
• "Asm table" an include file for AVR assembler is generated. 

Under Windows, the files are written by default to the same folder in which the exe
cutable resides. Under Linux Wine, the files are all written by default to the documents 
folder.



3.1 Scheme

The file names starts with "RMat_" and then includes 

• the variations StIn, SiIn, StEx or SiEx of the 
matrix, 

• the E row used, 
• the resistor tolerance, 
• the ADC resolution. 

In  that  case  the  name  of  the  graphic  file  is 
"Rmat_StIn_E12_1%_10bit.gif".

3.2 Resistor list

The resistor file holds the content of the Results-Resistors field as plain text.

3.3 Assembler table

The assembler table "filename.inc" looks like this: 

; C4R4 keypad, Inclusive, Stacked
; Decimal table for assembler
Keytable: ; lower limit, upper limit+1, 'Key'
.DW 20, 22 ; '1'
.DW 54, 57 ; '2'
.DW 66, 70 ; '3'
.DW 84, 88 ; 'A'
.DW 88, 92 ; '4'
.DW 214, 222 ; '5'
.DW 253, 262 ; '6'
.DW 304, 314 ; 'B'
.DW 316, 326 ; '7'
.DW 569, 580 ; '8'
.DW 623, 634 ; '9'
.DW 683, 693 ; 'C'
.DW 732, 741 ; '*'
.DW 896, 901 ; '0'
.DW 918, 923 ; '#'
.DW 940, 944 ; 'D'
.DW 0 ; table end
Keyvalues:
.DB "123A456B789C*0#D" ; Key values ASCII

It can be directly included in an assembler source code file.

Note that with 8 bit ADC selected the include file is different as the table is in bytes in
stead of words! 

4 Experiences made
The inclusive mode works very fine and approximation mostly ends with zero over
laps. If not, just restart the approximation.

The exclusive mode works fine if the number of keys is 12 or less. In case of 9 keys or 
more 1% tolerance or 10-bit ADC has to be selected (or both) to achieve successful 
results. With 16 keys most runs end with overlaps in this mode. In that case either 
stick with the default values or manually correct resistor values.

Stacked or single resistor mode results in different values, but shows no different ap
proximation behavior.  



AVR-Single-Chip-
Processors AT90S, 

ATtiny, ATmega
of ATMEL in practical examples.

12- and 16-pin keypad on an AVR
This page shows 

1. connecting a 12- or 16-key keypad to an AD converter input, 
2. how to calculate the resistor values, using provided software, 
3. how to optimize the assembler software to read the keypad. 

1 Connecting a keypad to an AD converter input

1.1 Connections, 12 keys

The scheme shows a 12 key wide keypad and its exter
nal  resistor  matrix.  The four  rows are connected with 
four stacked resistors that are tied to the positive oper
ating  voltage.  The  three  columns  are  connected  with 
three stacked resistors that are tied to the negative op
erating voltage.

If none of the keys is pressed, there is no connection be
tween plus and minus, the output voltage at point Y is 
zero because R1, R2 and R3 pulls it down to the nega
tive operating voltage. If Y is connected with an analog-
to-digital converter (ADC), its state reads as zero.

If a key is pressed, e.g. key 5, it connects the row line 
(here:  the  horizontal  connection  between  4-5-6)  with 
the column line (here the vertical connection between 2-
5-8-0). 

1.1.1 The resistors if key 1 is pressed

This is what happens if the key 1 is pressed: this key con
nects row 1 with column 1. Now R1, R4, R5, R6 and R7 op
erate as a voltage divider. The current through these five re
sistors in a row, I1 = U+ / (R1 + R4 + R5 + R6), causes a 
small voltage on R1, Y1 = I1 * R1. As the ADC input is of a 
very high resistance, the voltage on R1 appears on Y and the 
ADC sees this voltage.

The formula below combines the current calculation and the 
voltage calculation. The calculated Y1 is the fraction of the 
operating voltage (U+) that appears on the ADC, whenever 
key 1 is pressed. 



1.1.2 The resistors if key 9 is pressed

If the key 9 is pressed, the voltage divider works different. 
Now R1, R2 and R3 together are the lower resistor of the 
voltage divider.  The upper resistor of  the voltage divider 
are the resistors R6 and R7. The current now is

 I9 = U+ / (R1 + R2 + R3 + R5 + R6 + R79

The voltage on Y is then
Y9 = I9 * (R1 + R2 + R3).

Every key combines specific resistors to a voltage divider, 
and so produces a specific voltage on output Y. With that 
voltage it can be determined if any one and which specific 
key is currently pressed. The task then is to select the re
sistor values in an optimal way (using the software provid
ed below).

1.2 Keypad with 16 keys

Four more keys require an additional resistor to work in 
a similar way. Looks very much like the 12-key version, 
but the calculation of the voltage divider is different and 
produces very different voltages. In that case the resis
tors have to be of a higher accuracy of 2 or 1% (see 
below).

1.3 Alternative schematic with parallel resistors

The four upper resistors do not have to be stacked necessarily, they can also be paral
lel. The following schemes show this alternative.

This alternative schemes change the mode of calculation slightly, but the main calcula
tion scheme remains the same.

2 Calculating resistor values
Here a tool for calculating resistors is provided and introduced. 



2.1 Calculation basics and rules

This chapter shows the rules and how to calculate resistor values in a correct way. 
Calculating the resistors for those voltage dividers is not trivial: 

• Resistors are available with an accuracy of plus and (!) minus 0.1, 1, 2 or 5% 
only. Resistor values with 5% inaccuracy so are within a band of 10% of the 
nominal value. For a 1k resistor that means that its value can be anywhere be
tween 950 and 1,050 Ohms. The resulting voltages produced so are within a 
certain bandwidth. The minimum voltage produced appears if all resistors in the 
lower resistor chain are on their lowest value and all those in the upper resistor 
chain are in their upper bandwidth value. The maximum voltage is gained if the 
lower resistor chain is on their highest value and the resistors of the upper 
chain on their lowest. Those inaccuracies have to be taken into account, the 
bandwidth requires that a lower and upper limit has to be checked if the key is 
to be identified. With that we do not have to select accurate values manually 
and we do not need to keep the temperature of our matrix constant. Fortunate
ly we do not need resistors with tolerances of 0.1%, so we can forget this. 

• Resistors are only available with certain discrete values, known as E rows. The 
most common one is the E12 row: 1.0 - 1.2 - 1.5 - 1.8 - 2.2 - 2.7 - 3.3 - 3.9 -  
4.7 - 5.6 - 6.8 - 8.2, and their decadic manifolds, such as 10 - 12 - etc. or 100 
- 120 - etc.. The E48 row has double, the E96 row four times more discrete val
ues er decade. So we can only select resistor values from these rows. Solving 
the problem with discrete formulas (with 7 resp. 8 unknowns) would in almost 
every case lead to resistor values that are not commercially available, so forget 
this theoretical calculation option. 

• Even having those extended E48 and E96 rows, most of these resistors are not 
available commercially in practice. Most sellers fear having hundreds of boxes 
with exotic values only rarely sold and sitting around in their boxes for years. 
So we have to stick with the E12 and E24 row, and E48 or E96 values are only 
theoretical (or only for those with excellent connections to certain dealers). You 
can play around with those extended rows, but a visit in your local electronics 
store or a look into the catalog of your preferred parts dealer gets you down to 
earth and to E12 values. Finally the good message: the problem is resolvable 
with only E12 values.

• Whenever we change one resistor value to a different one in the E row, nearly 
the whole set of voltages (and their voltage bandwidth) changes. So selecting 
values has to be iterated step by step, keeping an ideal state in mind (the volt
ages produced are uniformly deployed over the whole range from 0 to the oper
ating voltage) and no overlaps occur (every key produces its unique voltage 
range). 

• Another criterion is that pressing keys does not lead to elevated currents, e.g. 
of 100 mA. The sum of the resistors R1 and R7 (resp. R8) so should not be 
smaller than 1,000 Ohms, causing a maximum current of less than 5 mA. 

• The other case would be too large resistor values. If the currents through the 
matrix would be less than 10µA the potential influence of noise would be too 
large. So all the resistor values should be divided then by ten reducing noise on 
the voltages. 

• If the AD converter works with the operating voltage of the keypad as reference 
voltage, the design is especially easy. If 8 bit resolution is selected (the left ad
just result bit ADLAR is set to one, only the MSB of the ADC is read), the results 
are between 0 and 255. In the 10 bit case the results are between 0 and 1023. 
In both cases no conversion to voltages is necessary. 



2.2 Calculation software

As designing 
such  matri
ces  is  a 
lengthy 
process,  I 
designed 
this  soft
ware tool. It 
allows  to 
play  around 
with  values 
and  param
eters.

The software provided here (Win64 executable) or here for Linux i386-x64 is a simple 
command line application. Download it,  unzip the executable and start it.  You can 
download the  pascal source code, too, compile it with the Free Pascal Compiler fpc 
and run this executable. If you are cautious with downloaded executables or if you 
need the executable file for other operating systems, you can follow this way.

The version including upper serial as well as parallel resistors is provided  as Win64 
executable here and as Free Pascal Source Code here.

The software allows to 

• select between 12 and 16 key keypads (switch by pressing the key k), 
• single step through the iteration process (press s for a single step, selects one 

of the resistors to be processed randomly), 
• perform 1,000 iteration steps at once (press i), 
• select 8 or 10 bit resolution of the ADC (press r, input 8 or 10), 
• set resistors R1 to R7 resp. R8 to desired values manually (press n, input resis

tor value in Ohms), 
• select resistor accuracy and E row (press t, input 1, 2 or 5% tolerance, input re

sistor row E24, E48 or E96), 
• write the current results to a text file (press w), 
• exit the application (press x). 

The window displays 

• the current resistor values R1 to R7 resp. R8, 
• the ADC results for the 12 resp. 16 keys, according to the current ADC resolu

tion, as nominal, minimum and maximum values as well as the ideal reference 
value, 

• the current key row by produced nominal voltages (can change), 
• the selected resistor row (E12, E24, E48, E96), 
• the  number  of  current  overlaps  (total  number  of  overlapping  bandwidths, 

should be zero), 
• the sum of differences between the nominal values and their ideal reference 

values (should be as small as possible), 
• the tolerance of the resistors and the ADC resolution. 

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resmatrix/resmatrix_parallel.pas
http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resmatrix/resmatrix_parallel_Win64.zip
http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resmatrix/resmatrix_parallel_Win64.zip
http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resmatrix/resmatrix.pas
http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resmatrix/resmatrix_lin.zip
http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resmatrix/resmatrix.zip


2.3 Example calculation 12 keys

The following example is based on the following parameters: tolerance 5%, 8 bit reso
lution, resistor row E12.

This is the result 
after  a  single 
step.  The  ran
dom step select
ed R4, decreased 
this resistor from 
68k  to  56k,  re
sulting  in  reduc
ing  the  overlaps 
and  the  differ
ences  between 
nominal and ref
erence  readings 
from 599 to 548. 

Another  step  in
creases  R5 from 
27k  to  33k,  re
ducing  overlaps, 
but  increasing 
the differences.

This is the result 
after  1,000 iter
ation  steps.  The 
number of over
laps is zero, the 
differences  are 
minimized to 32. 
All  voltages  are 
within  a  well 
suitable  band
width  of  ADC 
readings:  not 
too  small  band
width,  enough 
difference to the next key.

Note that the iteration step does not always lead to optimal results. As the steps are 
done randomly, different runs might lead to sub-optimal one-way-streets. Try another 
iteration (select 16 keys, then 12 keys again, then restart iteration) or change resis
tors manually.



2.4 Example calculation 16 keys

By  pressing  k 
the  calculation 
switches  to  16 
keys,  1%  toler
ance and 10 bits 
resolution of the 
ADC.  These  are 
necessary values 
to get the matrix 
working  (5%  is 
too  large,  2% 
would  also  be 
sufficient,  8  bit 
is  too  small). 
Resistor row still 
is  E12  to  ease 
availability.

With  standard 
resistors  the 
number of over
laps is already at 
zero,  but  differ
ences can be op
timized.

The  result  after 
1,000  iteration 
steps. Also looks 
fine. Problem solved. 

2.5 12 key result files

A  view  on  the  result  files  docu
menting the results, with a simple 
editor. The name of the result file 
reflects the parameters used, the 
example file for 12 keys is here.

The  upper  portion  of  the  files 
shows  the  program  output.  The 
lower  portion  provides  a  table  to 
be copied to the assembler source 
file. The table can be used to iden
tify  the key pressed (see below). 
The first value on every line holds 
the 8-bit lower limit for the respec
tive key, the second the value at 
the end of the bandwidth plus one. 
So a key is recognized if the ADC 
result is equal or above the lower 
limit  (a  CP  instruction  ends  with 
the carry flag clear) AND is smaller 

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resmatrix/RMatrix_12Keys_5%25_E24_8bit.txt


than the given upper limit plus one (a CP instruction ends with the carry flag set).

The table ends with two zero bytes to signal that the end of table has been reached.

The last two lines in the file provide an ASCII table with the respective key values. 

2.6 16 key result files

The  file  name  pro
duced reflects the pa
rameters  used,  the 
example  file  is  here. 
The  upper  portion  of 
the  file  again  shows 
the program output.

The  lower  portion 
again provides a table 
to  be  copied  to  the 
assembler source file. 
Its  content  are  now 
two 16-bit values per 
line,  with  the  lower 
limit  and  the  upper 
limit plus one.

The table ends with a 
zero  word  to  signal 
that the end of table 
has been reached.

The last  two lines  in 
the file again provide 
an  ASCII  table  with 
the  respective  key 
values. 

3 Assembler software
The assembler software is a little bit different for the 8 and 10 bit ADC versions. 

3.1 Assembler software for 8 bit ADC values

The following assumes 

• that the 8-bit ADC value is in register R2, 
• registers R0 and R1 are available and can be temporarily used, 
• register pair Z can be used. 

The table as provided in the result file is included: 

; Decimal table for assembler
Keytable: ; lower limit, upper limit +1), 'Key'
.DB 25, 31 ; '1'
.DB 34, 41 ; '2'
.DB 45, 54 ; '3'
.DB 66, 78 ; '4'
.DB 84, 96 ; '5'
.DB 104, 117 ; '6'

http://www.avr-asm-tutorial.net/avr_en/apps/key_matrix/keypad/resmatrix/RMatrix_16Keys_1%25_E24_10bit.txt


.DB 117, 130 ; '7'

.DB 138, 151 ; '8'

.DB 159, 171 ; '9'

.DB 178, 189 ; '*'

.DB 194, 204 ; '0'

.DB 209, 217 ; '#'

.DW 0 ; table end
Keyvalues:
.DB "123456789*0#" ; Key values ascii 

The following subroutine first reads in the lower limit from the table. If the lower limit 
is zero, the table is at the end and no key is recognized (routine returns with FF in 
R0). Then it checks if the ADC value is smaller then the lower limit. If yes, no key is 
pressed or the value is in between two valid key limits (routine returns with FF in R0). 
Then the upper limit plus one is read from the table. If the ADC value is higher than 
this upper limit, the routine continues from the start. If the ADC value is smaller, a 
valid key has been detected, the respective ASCII value is read from the table and re
turned in R0. 

GetKey12: ; get the ascii value of the key pressed, return result in R0
    clr R1 ; clear counter                        +-------------------+
    dec R1 ; set counter to FF                    | Set counter R1=FF |
    ldi ZH,HIGH(2*Keytable) ; point Z to table    | Set Z to table    |
    ldi ZL,LOW(2*Keytable) ;                      +-------------------+
GetKey1:                                                    |
    lpm ; read first value (lower limit)             ->Read lower limit
    tst R0 ; table at the end?                      /      / \__________> yes
    breq GetKey3 ; table end reached               |       \0/no         |
    inc R1 ; count up                              |    Inc counter      v
    cp R2,R0 ; compare ADC result with lower limit |       / \__________>|
    brcs GetKey3 ; smaller than lower limit        |       \C/no         |
    adiw ZL,1 ; upper limit in table               |  Read upper limit   |
    lpm ; read to R0                               |    and compare      |
    adiw ZL,1 ; next byte                          |        |            |
    cp R2,R0 ; compare with upper limit             \______/ \           |
    brcc GetKey1 ; larger than upper limit               no\C/           |          
    ldi ZH,HIGH(2*KeyValues) ; point to result table  Z to value table   |
    ldi ZL,LOW(2*KeyValues) ;                               |            |
    add ZL,R1 ; add counter                           Add counter        |
    brcc GetKey2 ; no overflow                              |            |
    inc ZH ; overflow, add one to MSB                 Inc MSB table      |
GetKey2: ;                                                  |            |
    lpm ; read character to R0                         Read char to R0   |
    ret ; return with result                               RET           |
GetKey3: ; table at end or value in between, return FF                   v
    clr R0 ; clear result                                R0 to FF <------
    dec R0 ; return FF                                      |
    ret ;                                                  RET

That is it all. 

3.2 Assembler software for 10 bit ADC values

10 bit ADC values require word-wise comparison. The following registers are used: 

• R0 returns the pressed key character, if invalid the FF, 
• R2:R1 is used for the table value, 
• R4:R3 stores the input value, 
• R5 counts, 
• Z is used for pointer purposes. 

; Decimal table for assembler, 16 keys
Keytable: ; Lower lim, upper lim+1, 'Key'
.DW 96, 100 ; '1'
.DW 122, 127 ; '2'
.DW 158, 164 ; '3'
.DW 191, 198 ; 'A'
.DW 263, 272 ; '4'
.DW 319, 329 ; '5'



.DW 388, 399 ; '6'

.DW 444, 455 ; 'B'

.DW 497, 508 ; '7'

.DW 566, 578 ; '8'

.DW 640, 650 ; '9'

.DW 693, 703 ; 'C'

.DW 742, 751 ; '*'

.DW 794, 802 ; '0'

.DW 842, 849 ; '#'

.DW 874, 880 ; 'D'

.DW 0 ; table end
Keyvalues16:
.DB "123A456B789C*0#D" ; Key values ascii 
;
; Convert ADC value in R4:R3 to key ascii code
GetKey16:
        clr R5 ; clear counter
        dec R5 ; set to FF
        ldi ZH,HIGH(2*Keytable) ; Z to table
        ldi ZL,LOW(2*Keytable)
GetKey16a:
        inc R5 ; increase counter
        lpm ; read LSB from table
        adiw ZL,1 ; next byte
        mov R1,R0 ; copy to R1
        lpm ; read MSB from table
        adiw ZL,1 ; next byte
        mov R2,R0 ; copy to R2
        tst R1 ; LSB zero?
        brne GetKey16b ; no, go on
        tst R2 ; MSB zero
        breq GetKey16d ; table at end
GetKey16b:
        cp R3,R1 ; compare LSB
        cpc R4,R2 ; compare MSB plus overflow
        brcs GetKey16d ; carry indicates value smaller then lower limit
        lpm ; read LSB upper limit
        adiw ZL,1 ; next byte
        mov R1,R0 ; copy to R1
        lpm ; read MSB upper limit
        adiw ZL,1 ; next byte
        mov R2,R0 ; copy to R2
        cp R3,R1 ; compare LSB
        cpc R4,R2 ; compare MSB plus overflow
        brcc GetKey16a ; ADC value higher or equal upper limit
        ldi ZH,HIGH(2*KeyValues16) ; Z to table
        ldi ZL,LOW(2*KeyValues16)
        add ZL,R5 ; add counter
        brcc GetKey16c ; no overflow
        inc ZH ; overflow, add 1 to MSB
GetKey16c:
        lpm ; read ascii value from table to R0
        ret ; return
GetKey16d:
        clr R0 ; set R0 to FF
        dec R0
        ret ; return with R0=FF

Hint: The LPM instructions used here are AVR old-style. By use of LPM Z+,R a few 
lines can be saved.

